```
1 '-------------------------------------------------------------------------------
2 'CALCULATION OF MARKOV EQUATIONS USING EXPLICIT OR IMPLICIT FINITE DIFFERENCE METHODS
3 '-------------------------------------------------------------------------------
4 CLS
5 Print
6 Print "  ____________________________________________________________________________"
7 Print
8 Print "  MARKOV-2 (MARKOV2.BAS).  MARKOV SOLUTION OF MULTI-STATE SYSTEM PROBABILITIES"
9 Print "  ORIGINAL AUTHORS: S. MASSOUD AZIZI AND PHIL RUTHERFORD"
10 Print "  REVISION AUTHOR: PHIL RTHERFORD"
11 Print "  PAPER: http://www.philrutherford.com/Conf_Papers/markov.pdf"
12 Print "  ORIGINAL REVISION DATE: 10-12-1989"
13 Print "  CURRENT REVISION DATE:  11-09-2020"
14 Print "  RUN DATE: ";Date$;
15 Print " (DD-MM-YYYY)
16 Print "  RUN TIME: ";TIME$
17 Print "  RUN WITH MMBASIC.EXE (www.mmbasic.com)"
18 Print
19 Print "  COMPUTES TIME-DEPENDENT, MULTI-STATE PROBABILITIES USING EXPLICIT OR "
20 Print "  IMPLICIT FINITE DIFFERENCE METHODOLOGY.  THE SAMPLE PROBLEM USES DATA FOR A "
21 Print "  DEISEL GENERATOR TO COMPUTE PROBABILITIES OF BEING IN SIX STATES, STANDBY, "
22 Print "  OPERATION, TEST, PREVENTIVE MAINTENANCE, FAILURE AND REPAIR."
23 Print
24 Print "  IF DESIRED, USER CAN EDIT STATE TRANSITION FREQUENCIES VIA THE DATA "
25 Print "  STATEMENTS AT PROGRAM END. SEE PAPER FOR DERIVATION OF TRANSITION "
26 Print "  FREQUENCIES."
27 Print
28 Print "  THE IMPLICIT METHOD IS THE PREFERABLE OPTION. THE EXPLICIT METHOD REQUIRES "
29 Print "  SHORTER TIME STEPS. A TIME STEP OF 4 HRS OR LESS GIVES STABLE RESULTS THAT "
30 Print "  COMPARE WELL TO THE IMPLICIT METHOD. A TIME STEP OF 5 HRS OR MORE IS "
31 Print "  UNSTABLE."
32 Print
33 Input "  ENTER INITIAL TIME (HRS)           ";T
34 Input "  ENTER FINAL TIME (HRS)             ";TMAX
35 Input "  ENTER TIME STEP (HRS)              ";DELT
36 READ IMAX           'NO. OF STATES
37 JMAX = IMAX
38 Print "  NUMBER OF STATES                   ",IMAX
39 Input "  ENTER IMPLICIT (I) OR EXPLICIT (E) ";OPT$
40 Print

41
42 If OPT$ = "I" Or OPT$ = "i" Then Print "  *** IMPLICIT FINITE DIFFERENCE OPTION ***"
43 If OPT$ = "E" Or OPT$ = "e" Then Print "  *** EXPLICIT FINITE DIFFERENCE OPTION ***"
44 Print

45
46 Print "                 *** LIST OF TRANSITION FREQUENCIES (/HR) *** "
47 Print
48 Dim ST$(IMAX)
49 Dim LAM(IMAX,JMAX)
50 Dim P(IMAX)
51 Dim Q(IMAX)
52 Dim A(IMAX,JMAX+1)
53 Dim X(IMAX)
54 Dim B(IMAX)
55 Dim S(IMAX)
56
57 100 For I=1 To IMAX
58 110 READ ST$(I)                   'SYSTEM STATE DESCRIPTIONS
59 120 Print TAB(I*11+2); ST$(I);
60 125 Next I
61
62 130 For I=1 To IMAX
63 135 Print TAB(I*11+2); "---------";
64 140 Next I
65 145 Print
66
```

```
150 For I=1 To IMAX
155 For J=1 To JMAX
160 READ LAM(I,J)
165 Next J
170 Next I

175 FOR I=1 TO IMAX
180 Print TAB(3);ST$(I);
185 For J=1 To JMAX
190 Print TAB(J*11+2);
195 Print LAM(I,J);
200 Next J
210 Print
215 Next I

220 Print
225 Print "                 *** RESULTANT SYSTEM STATE PROBABILITIES ***"
230 Print

235 For I=1 To IMAX
240 Print TAB(I*16-5); ST$(I);
245 Next I

250 For I=1 To IMAX
255 Print TAB(I*16-5); "---------";
260 Next I

265 Print
270 Print "  T(HR)";

275 For I=1 To IMAX
280 Print TAB(I*16-5); " Pr("I")";
285 Next I
286 Print TAB((I)*16-8);
287 Print "Checksum"

290 Print
295 Print TAB(2);T;

300 For I=1 To IMAX
305 READ P(I)                       'INITIAL STATE PROBABILITIES
310 Print TAB(I*16-8);
315 Print P(I);
317 CHKSUM = CHKSUM + P(I)
320 Next I
321 Print TAB((I)*16-6);
322 Print CHKSUM

330 T=T+DELT                        'CALCULATE AND PRINT TIME STEPS
335 Print TAB(2);T;

340 SUM=0                  'SUM IS USED IN CONJ. WITH 810-860 TO NORMALIZE PROBABILITIES
345 If OPT$ = "I" Or OPT$ = "i" Then Goto 420

350 FOR I=1 TO IMAX                 'EXPLICIT METHOD CALCULATION
355 SOURCE=0
360 SINK=0
365 FOR J=1 TO JMAX
370 SOURCE=SOURCE + LAM(J,I)*P(J)
375 SINK=SINK + LAM(I,J)*P(I)
380 NEXT J
385 P(I)=P(I)+(SOURCE-SINK)*DELT
390 NEXT I
395 Goto 810

420 For I=1 To IMAX                 'IMPLICIT METHOD CALCULATION
```

```
430 Q(I)=O
435 FOR J=1 TO JMAX
440 Q(I)=Q(I)+LAM(I,J)              'CALCULATE LHS OF IMPLICIT EQUATION
445 NEXT J
450 FOR J=1 TO JMAX
455 A(I,J)= -DELT*LAM(J,I)/2
460 IF I=J THEN A(I,J)= 1+DELT*Q(I)/2
465 NEXT J
470 X(I)=0
475 FOR J=1 TO JMAX
480 X(I)= X(I)+DELT*LAM(J,I)*P(J)/2  'CALCULATE RHS OF IMPLICIT EQUATION
485 NEXT J
490 B(I)=(1-DELT*Q(I)/2)*P(I)+X(I)
495 A(I,JMAX+1)=B(I)
500 Next I

550 FOR K=1 TO IMAX-1               'PROCESS OF GAUSSIAN ELIMINATION
560 FOR I=K+1 TO IMAX
565 FACTOR=A(I,K)/A(K,K)
570 FOR J=K TO JMAX+1
580 A(I,J)=A(I,J)-FACTOR*A(K,J)
590 NEXT J
600 NEXT I
610 Next K

700 FOR I=IMAX TO 1 STEP -1         'BACK-SUBSTITUTION CALCULATION
710 S(I)=0
720 FOR K=IMAX TO I+1 STEP -1
730 S(I)=S(I)+A(I,K)*P(K)
735 IF K=I THEN S(I)=0
740 NEXT K
750 P(I)=(A(I,JMAX+1)-S(I))/A(I,I)
760 Next I

810 For I=1 To IMAX
820 SUM=SUM+P(I)                    'NORMALIZATION OF P(I)
830 NEXT I
840 FOR I=1 TO IMAX
850 P(I)=P(I)/SUM
860 Next I

890 CHKSUM = 0
900 For I=1 To IMAX                 'PRINT RESULTS
910 Print TAB(I*16-8);
920 Print P(I);
925 CHKSUM = CHKSUM + P(I)
930 Next I
932 Print TAB((I)*16-6);
934 Print CHKSUM

950 If T = TMAX Then Goto 1085
960 Goto 330

1085 Print
1090 Print "  ____________________________________________________________________________"
1100 End

1500 DATA 6
1700 DATA " STANDBY", "OPERATION", "  TEST", "  P.M.", " FAILURE", " REPAIR"
1800 DATA 0.00E+1, 1.14E-5, 1.37E-3, 6.85E-4, 3.00E-4, 0.00E+1
1900 DATA 4.20E-2, 0.00E+1, 0.00E+1, 0.00E+1, 3.00E-3, 0.00E+1
2000 DATA 4.50E-1, 1.10E-5, 0.00E+1, 3.50E-2, 1.50E-2, 0.00E+1
2100 DATA 3.10E-2, 1.10E-6, 2.70E-2, 0.00E+1, 3.10E-3, 1.25E-3
2200 DATA 0.00E+1, 0.00E+1, 0.00E+1, 0.00E+1, 0.00E+1, 2.50E-1
2300 DATA 7.50E-2, 0.00E+1, 4.38E-2, 0.00E+1, 6.25E-3, 0.00E+1
2400 DATA 1,0,0,0,0,0
```